

# Language Reference Manual for Propeller

## A Property Manipulation Language

Isra Ali

*Tester*

`isra.hamid.ali@tufts.edu`

Gwendolyn Edgar

*Manager*

`gwendolynedgar@tufts.edu`

Randy Price

*Language Guru*

`edward.price@tufts.edu`

Chris Xiong

*System Architect*

`lxiong01@tufts.edu`

February 28, 2022

## 1 Overview

Propeller is a language designed to write programs using the reactive programming paradigm. Programs written in Propeller usually act upon external events such as user inputs and environmental changes. It has a binding system: functions can be bound to properties of objects (akin to members or fields in other languages), so that they get called whenever the value of these properties change.

Propeller has several possible runtime environments: a simple runtime with a fixed entry point, one with GUI capabilities, and one that monitors certain local parameters of the computer (e.g. CPU temperature sensor readings, files creation and deletion, load average and so on).

As of now, the syntax and semantics described in this document is tentative. The final version may contain revisions to address unforeseen design flaws.

### 1.1 Notation used in this document

This document uses a variation of Backus-Naur form to describe the syntax of the language. The most significant additions are the character range notation ("**a**"-"**d**" rather than "**a**" | "**b**" | "**c**" | "**d**"), and items repeated zero or more times (**{<characters>}** means **<characters>** repeated zero or more times).

## 2 Syntax

### 2.1 Comment

Anything following a hash (#) that is not part of a string literal will be treated as a comment.

### 2.2 Literals

```
<digit> ::= "0"-"9"
<int-literal> ::= {<digit>}
<float-literal> ::= {<digit>} "." {<digit>}
<bool-literal> ::= "true" | "false"
<string-characters> ::= character or escape sequence
<string-literal> ::= "'" {<string-characters>} "'"
```

For strings, escape sequences are interpreted the same way OCaml interprets them (as in `Scanf.unescaped`).

### 2.3 Identifiers

```
<letter> ::= "A"-"Z" | "a"-"z"
<identifier> ::= <letter> ("?" | "")
                | <letter> ("_" | "")
                {(<letter> | <digit>) | ((<letter> | <digit>) "_")
                (<letter> | <digit>) ("?" | "")}
```

To put this rather cryptic identifier rule in plain English: only letters, digits, underscores and question mark are allowed in identifier names. They must start with a letter, and must not end with an underscore. The question mark can only appear at the end, and two punctuation marks cannot appear consecutively.

### 2.4 Expressions

```
<binary-operators> ::= "+" | "-" | "*" | "/" | "%"
                    | "==" | "!=" | "<" | "<=" | ">" | ">="
                    | "and" | "xor" | "or"
<unary-operators> ::= "-" | "not"
<expr-list> ::= <expression> | <expr-list> "," <expression>
<arg-list> ::= "" | <expr-list>
<expression> ::= <int-literal>
                | <float-literal>
                | <bool-literal>
                | <string-literal>
                | <list>
                | <identifier>
                | <expression> <binary-operators> <expression>
                | <unary-operators> <expression>
```

```

| <identifier> "=" <expression>
| "[" <identifier> "]"
| "(" <expression> ")"
| <identifier> "(" <arg-list> ")"
| <identifier> "." <identifier>
| <identifier> "." <identifier> = <expression>

```

Operator precedence follows normal conventions.

## 2.5 List Literals

```
<list> ::= "[" arg-list "]"
```

## 2.6 Declarations

### 2.6.1 Variable Declaration

```

<types> ::= "int" | "float" | "bool" | "str" | "void" | <identifier> | <types> "&#x26;"
<variable-decl> ::= <types> <identifier> ";&#x26;"
<variable-decl-list> ::= "" | <variable-decl-list> <variable-decl>

```

### 2.6.2 Function Declaration

```

<formal-list> ::= <types> <identifier>
                | <formal-list> "," <types> <identifier>
<formal-list-opt> ::= "" | <formal-list>
<function-decl> ::= "fn" <identifier> "(" <formal-list-opt> ")" "->"
                  <types> "{" <variable-decl-list>
                  <statement-list> "}"

```

### 2.6.3 Object Type Declaration

```

<obj-def> ::= "objdef" <identifier> "{" <variable-decl-list> "}"
<external-obj-def> ::= "external" "objdef" <identifier> "{" <variable-decl-list> "}"

```

## 2.7 Statements

### 2.7.1 Sequencing

```

<statement-list> ::= "" | <statement-list> <statement>
<statement-block> ::= "{" <statement-list> "}"

```

### 2.7.2 Control Flow

- Branching

```

<if-statement> ::= "if" <expression> <statement-block>
                <elif-statements> <else-clause>
<elif-statements> ::= {"elif" <expression> <statement-block>}

```

```
<else-clause> ::= "" | "else" <statement-block>
```

Else clauses are attached to the closest unmatched if clause before it.

- Loops

```
<for-statement> ::= "for" <identifier> "from" <expression>
                    "to" <expression> <statement-block>
<while-statement> ::= "while" <expression> <statement-block>
```

- Jumps

```
<break-statement> ::= "break" ";"
<continue-statement> ::= "continue" ";"
<return-statement> ::= "return" (" " | <expression>) ";"
```

### 2.7.3 Special Statements

Bind and unbind has special syntactical rules. These rules exist only to simplify the implementation. For users of Propeller they can be treated as if they are normal built-in functions.

```
<bind-statement> ::= "bind" "(" <identifier> "." <identifier> ","
                    <identifier> ")" ";"
<unbind-statement> ::= "unbind" "(" <identifier> "." <identifier> ","
                    <identifier> ")" ";"
```

### 2.7.4 All Statements

```
<statement> ::= <expression> ";"
              | <if-statement>
              | <for-statement>
              | <while-statement>
              | <break-statement>
              | <continue-statement>
              | <return-statement>
              | <bind-statement>
              | <unbind-statement>
```

### 3 Semantics

Propeller is heavily inspired by languages in the C family, including its semantics. For the sake of brevity, semantics that are commonly found in other widely-adopted languages will be omitted.

#### 3.1 Data Types

Propeller has 5 primitive types: `int` is a signed 32-bit integer type; `float` is a 64-bit IEEE 754 floating point type; `bool` is a boolean type, stored in a single byte; `str` is a syntactic sugar for integer lists, which stores UTF-8 encoded characters of a string; and finally a `void` type.

Propeller has a list type. Lists are immutable. All items in one list must be of the same type.

Custom types can be defined. They can have a number of properties in them. A runtime can have several predefined custom types, which still must be declared in the program before use, but prefixed by a `external` keyword.

#### 3.2 Binding

Property of objects can be bound to functions, so that whenever the value of this property changes, functions bound to that property are called.

Let  $\beta$  be the bindings that are currently established during execution of the program.  $\beta$  is one of the environment metavariable of Propeller's operational semantics.  $\beta(o, p)$  is a set of functions bound to property  $p$  of object  $o$ . Note that this way objects of the same custom type don't share bindings.

Function bound to a property must accept 3 parameters: two values of the same type as the property itself, passing the old value of the property and new value of the property, and one of the object's type, which will be set to the object whose property value has changed.

When multiple functions are bound to the same property of an object, their order of execution is not defined.

Semi-formal operational semantics of syntactical forms related to binding will be given below.  $\rho(o, p)$  retrieves the location where property  $p$  of object  $o$  is stored, and  $\sigma(l)$  is the value at location  $l$ .

$$\begin{array}{c}
\langle e, \rho, \sigma, \beta, \dots \rangle \Downarrow \langle v, \rho, \sigma_0, \beta, \dots \rangle \\
\text{for each } f_i \in \beta(o, p), i = 1 \dots n \\
\frac{\langle f_i(\sigma_0(\rho(o, p), v, o), \rho, \sigma_{i-1}, \beta, \dots) \Downarrow \langle \text{void}, \rho, \sigma_i, \beta, \dots \rangle}{\langle \text{PROPERTYASSIGN}(o, p, e), \rho, \sigma, \beta, \dots \rangle \Downarrow \langle v, \rho, \sigma_n\{\rho(o, p) \mapsto v\}, \beta, \dots \rangle} \quad \text{PROPERTYASSIGN} \\
\\
\frac{}{\langle \text{BIND}(o, p, f), \rho, \sigma, \beta, \dots \rangle \Downarrow \langle \text{void}, \rho, \sigma, \beta\{(o, p) \mapsto \beta(o, p) \cup \{f\}\}, \dots \rangle} \quad \text{BIND} \\
\\
\frac{}{\langle \text{UNBIND}(o, p, f), \rho, \sigma, \beta, \dots \rangle \Downarrow \langle \text{void}, \rho, \sigma, \beta\{(o, p) \mapsto \beta(o, p) \setminus \{f\}\}, \dots \rangle} \quad \text{UNBIND}
\end{array}$$

(Note: If the object is of an external type in PROPERTYASSIGN, the actual behavior will differ a little, but that is the case only due to technical reasons and has no difference semantically.)

### 3.3 Program Execution

When a program written in Propeller is executed, it will start from a function called `init()`. After `init()` returns, it enters an event loop defined by the runtime library. For the most basic text-mode only runtime, the event loop simply terminates the program.

## 4 Built-in Functions and Standard Library

This section gives a tentative feature set of built-in functions and the standard library of Propeller.

Functions that are built-in will offer mathematical calculations such as trigonometric functions, terminating execution with exit code, printing values of primitive types, as well as basic list operations.

The standard library, built upon these built-in functions, will provide extra features such as string manipulation, printing values in lists, and some more advanced list operations. The standard library will be implemented using Propeller itself, and included implicitly in every program.

## 5 Examples

### 5.1 Minimal test driving program

This is a basic text mode program. It should print 3 and then terminate. It uses the basic text mode runtime.

```
1 # a simple custom type containing one int property
2 objdef A
3 {
4     int prop;
5 }
6
7 A a;
8
9 fn print_prop(int oldval, int val, obj o) -> void
10 {
11     # built-in function
12     print_int(val + 2);
13     # function in standard library to terminate execution
14     exit(0);
15 }
16
17 fn init() -> void
18 {
19     a.prop = 0;
20     bind(a.prop, print_prop)
21     # now that a.prop has been modified, print_prop will be called.
22     a.prop = 1;
23 }
```

### 5.2 Temperature Monitor

A simple temperature monitor that prints a message when the reading exceeds a threshold. This program will require a correct runtime.

```
1 external objdef Sensor
2 {
3     float temperature;
4 }
5
6 Sensor sensor;
7
8 fn print_warning(float t) -> void
9 {
10     if (t > 100)
11     {
12         print_str('Boiling!');
13     }
14 }
15
16 fn init() -> void
17 {
18     bind(sensor.temperature, print_warning);
19 }
```